



The Weaving Turtle • African Textiles

by

Orlando Mihich

Joan of Arc Junior High School
Community School District 3
New York City



© 1993 Orlando Mihich

You may copy and distribute this document for educational purposes provided that you do not charge for such copies and that this copyright notice is reproduced in full.

Acknowledgments

Thanks to Michael Tempel, Eadie Adamson, Glenderlyn Johnson, and Maynard Banks for their comments, encouragement, and advice. Special thanks to my students, Myckele Spencer, Michael Hailstock, Coedell Page, Henry Polanco, Antoine Draughon, and Michael Toribio for their exemplary work and dedication.

About this Document

The work described in this paper was originally created using GS LogoWriter on Apple IIGS computers. The programs were transferred to Macintosh LogoWriter so that the designs could be incorporated into this document, which is in Microsoft Word 5.0. The original drawings were in color. For this publication, many of them were redrawn in black and white, with variations in shading substituting for differences in color.

Introduction

Each school year I look forward to meeting my old and new students. I know that during the year I will involve my students, and myself, in new and meaningful learning experiences, and hopefully explore new areas. I also know that my students will enrich my world with their own ideas and concepts. It is always a two way process. In addition, I try to complete the year with a meaningful event in the lives of my students, in my professional life, and possibly with a contribution to the society in which we live and operate. Doing otherwise would be a wasted year.

Past Logo Projects

LogoWriter has been the ideal medium to involve students with projects ranging from Global Warming to AIDS, from the Operation of a Waste Water Treatment Plant to the Depletion of the Ozone Layer. Where feasible, students set up an experimental part to collect and use data. For the Depletion of the Ozone Layer project, for example, in addition to a complete LogoWriter program, students grew corn and bean plants under ultraviolet light and analyzed the collected results. For the past eight years, my students entered their projects in city-wide science fairs, and won many first prizes and honorable mentions. The idea of competing on this level is that students are enticed to work hard and to give the best of themselves. In return, they feel very good about themselves, and get a first hand experience in dealing with the competitive society they are about to enter.

The African Textile Project

I work at a Junior High School on the Upper West Side of Manhattan. The student population is exclusively Black and Hispanic. As part of the ninth grade curriculum I regularly introduce map making. This year, students created the map of the United States and other maps including that of the African Continent. Students included some information on old African Kingdoms and African culture in general. This work was orchestrated with Social Studies teachers Erin Kennedy and Dr. Elana Elster. Most of the information came from *Africa in History* (Davidson, Basil, *Africa in History*, MacMillan Publishing Co., NY, 1991). I continued with work in perspective and tessellations, among other subjects. African textiles, because of their very creative use of geometrical design and color, came as a normal extension in our tessellating work. I visited the market on 125th Street and brought back to class some samples of textile and a tie for myself. At this time, African textiles were very popular among some members of the staff and students who were wearing colorful caps and shirts with African motifs. The subject was, therefore, "in" and meaningful to my students.

In the past, I bought all the books I needed on African subjects at a nearby bookstore, Black Books Plus. Here I found several colorful postcards with reproductions of African textiles, and the well documented book *African Textiles* (Picton, John and Mack, John, *African Textiles*, Harper & Row, NY, 1989). I brought these, and other books to class and immediately some of my students started to write programs and

recreate the work. It became a class project that lasted for almost two months. The work was done on the Apple IIs using GS LogoWriter. The screens were quite beautiful.

It became clear to me that the screens should be seen by a broader audience. Properly printed and framed, they could and should be exhibited. There was no better place than in the community where the school is located. The intention was to give an opportunity to the students to exhibit their work in the community, and have the community meet the students and see the work. I discussed the idea with my students and several of them enthusiastically volunteered to work during after school hours to create additional screens for a possible exhibition. Michael Toribio, an ex-student of mine who went on to LaGuardia High School, and a Logo lover, joined us as project assistant.

We printed two of the initial screens, framed them in Lucite boxes and took them to the Black Books Plus bookstore. Glenderyn Johnson, the store owner liked them so much that she exhibited them in the store's window. We discussed the possibility of having an exhibition in the store. Ms. Johnson agreed immediately and we set a date. Prior to the exhibit, three students presented some of the screens at a regular meeting of the New York Logo Users Group. The reception was favorable, and the students felt they were on the right track.

Students created a flyer to be displayed in local stores to announce the exhibit, printed and mailed out invitations, generated a pamphlet to be given out at the exhibition, and prepared a guest book.

At the opening, in addition to the customary wall-hung work, two Apple IIs computers were showing a sequence of African textiles in vivid, colorful screens. Students were on hand to explain their LogoWriter programs to the visiting public; flipping the pages showing their procedures and shapes. Many visitors said that they learned something new from the kids that day. They praised the work as "illuminating" and of "excellent quality." The students interviewed the visiting public, added music to the LogoWriter screens, and created an "official" VCR tape of the event. During the opening, unexpectedly, they sold 21 of their creations. My students were once again winners. Proud of their work and of themselves they went on to a well deserved summer vacation.

African Textile

African textiles, as African masks, figures, pottery, and body ornaments and painting are all expressions of the rich African cultural heritage. These creations of art, through structural relationships of form, design, and colors express feelings that cannot be voiced any other way. Picasso, influenced by African art was capable to break the western figure and start a revolution in western art. He can be considered the first western genius coming out of Africa. Modigliani and Giacometti also created work deeply influenced by the art of Africa. The sculptures of the Dogon artist and


Giacometti are strong creations which speak the same universal language.

The word textile, comes from the Latin "texere" meaning to weave. African textile is rich in colors, and is endlessly creative in geometrical design. Handmade textile, from the hands of weavers, spinners, and dyers, is related to ancient pre-colonial traditions. Particular colors, shapes, and decorations may have political and ritual significance. Textiles are used not only to make articles of clothing, but also for decoration, to dress a house or a shrine during important events, or simply as gifts. The meaning of color varies from place to place. In Benin the color red is part of their ceremonial court dress, among the Ebira red color is associated with success, while in Madagascar red is applied to burial cloths. In most of West Africa, Ethiopia, East Africa, and Zaire all weaving is done by men. In North Africa and Madagascar, all weaving is done by women. In other areas both men and women weave. The most common materials used in weaving are wool, silk, cotton, bast, and raphia. Indigo is the most common dye, and is obtained from various plants of the genus *Indigofera*. Indigo gives various shades from pale blue to deep black. Other colors like red, yellow, blue, green, brown, and black are available from local vegetable and mineral sources.

Logo Textile


Islamic tessellations, Maya hieroglyphs and weaving, African masks and textiles are art forms that speak directly to the Logo programmer. It is impossible to visit a Mayan site without mentally creating at least a simple procedure to create and repeat the design. The same happened to me and my students with African textiles. My students taught the turtle to "weave." They did not simply copy the designs, but used their own color schemes, "stitched" the various parts of the design in their own way, and changed backgrounds to see the effect of different "materials."

The following screens are reflections not only of artistic and aesthetic feelings but also of knowledge of mathematics and geometry.



Embroidered raphia textile from Kuba, Zaire.


Myckele Spencer




Woolen textile from Niger.
Michael Toribio
Coedell Page




Cloth woven by the Asante people of Ghana.
Michael Hailstock



Cloth woven in Merina, Madagascar for aristocratic use.
Antoinne Draughton




Cotton Textile from Bamana, Mali
Myckele Spencer and Antoine Draughton




Cotton Textile
Yoruba, Nigeria
Myckele Spencer




Kente cloth woven by the Asante people of Ghana.
Michael Hailstock





Ceremonial Kente cloth from Ghana.
Myckele Spencer




Cloth used in men's ritual celebrations.
Woven by the Mende people of Sierra Leone.
Michael Hailstock



Cotton textile from Ebira, Nigeria
Myckele Spencer





Njave cloth from Mende, Sierra Leone.
Coedell Page

Logo Techniques

The African textile designs, with all their complexity, were drawn using variations of a few simple Logo graphics techniques.

The repetitious designs lend themselves to a procedural approach which is expressed in Logo.

Drawing and Filling

The red, yellow, and black Ghanaian textile shown at the top of page 5 was reproduced by Michael Hailstock. It is comprised of ten sections. Five of these sections are made up of a pattern of yellow and black rectangles, drawn by the procedure **cross**. The other five are made up of yellow and red trapezoids drawn by the procedure **arrows**.

The entire design is drawn by **textile**.

to textile

rg ht

pu setpos [-120 0]

pd

repeat 2 [arrows pu

rt 90 fd 60

lt 90 pd

cross pu

rt 90 fd 90

lt 90 pd]

Reset the graphics and hide the turtle.

Pick up the pen and place the turtle near the left side of the screen.

Put the pen down, ready to draw.

Draw **arrows**.

Turn and move to the right.

Turn to face towards the top of the screen.

Draw **cross**.

Turn and move to the right.

Turn to face towards the top of the screen.

The instructions within the brackets are done again because of **repeat 2**. Now there are four panels.

```
pd arrows
pu setpos [-150 -75]
seth 0 pd
repeat 2[cross pu
  rt 90 fd 90
  lt 90 pd
  arrows pu
  rt 90 fd 60
  lt 90 pd]
pd cross
end
```

Draw **arrows** for the fifth panel.

Pick up the pen and place the turtle near the left side of the screen, but 75 steps lower and 30 steps further to the left than before.

Aim straight up and put down the pen.

This time draw **cross** first, then **arrows**, and **repeat 2**.

Draw **cross** one more time.

The details are to be found in the procedures **arrows** and **cross**. **Arrows** calls the subprocedures **arrow.l** and **arrow.r** which draws right handed and left handed versions of the same pattern. The rest of **arrows** involves moving the turtle so that these two shapes will be drawn in the correct positions.

```
to arrows
arrow.l 60 14
arrow.r 60 12
lt 90
pu fd 30
rt 90 pd
arrow.r 60 12
rt 90 pu fd 60
lt 90 pd
arrow.l 60 14
pu fd 75
lt 90 fd 15
lt 90 pd
arrow.l 60 12
arrow.r 60 14
rt 90 pu fd 30
lt 90 pd
arrow.l 60 12
arrow.r 60 14
rt 90 pu fd 15
```

```
rt 90 bk 75
end
```

Arrow.l and **arrow.r** each take two inputs. The first is named *s* for size. The second is named *c* for color. Let's look more closely at **arrow.l**.

to arrow.l :s :c	
setc :c	Set the color.
fd :s	Go forward forming the long side of the arrow.
make "a pos	Remember where we are. This is the apex of the arrow.
bk :s	Go back to where we started.
lt 90	
fd :s / 4	Draw the base of the arrow.
rt 90	
fd :s / 4	Draw the short vertical side.
setpos :a	Complete the arrow by going to the apex, which we remembered earlier with make "a pos .
bk :s	Go back to the starting point.
lt 45	Point towards the upper left.
pu fd :s / 4	Pick the pen up and move to a point within the arrow.
pd fill	Fill the arrow in whatever color the turtle is.
pu bk :s / 4	Go back to the starting point.
rt 45 pd	Straighten up.
end	

Cross is built out of small rectangles, each drawn with the procedure **rec**. The two inputs to **rec** are the lengths of the sides, in all cases, 15 and 12. Unlike **arrow.l** and **arrow.r**, **rec** does not take an input which determines color. The **setc** command is used outside **rec**.

to cross	
setc 14	Set the color to yellow.
rec 15 12	Draw a rectangle.
pu fd 15 pd	Move to a new position.
setc 13	Set another color.

```

repeat 3 [rec 15 12
          pu fd 15 pd]
setc 14
rec 15 12
pu bk 60
rt 90 fd 12 lt 90
setc 13 pd rec 15 12
pu fd 15 pd
setc 14 rec 15 12
pu fd 15 pd
setc 13 rec 15 12
pu fd 15 pd
setc 14 rec 15 12
pu fd 15 pd
setc 13 rec 15 12
pu bk 60
rt 90 fd 12 lt 90 pd
repeat 2 [setc 13
          rec 15 12
          pu fd 15 pd]
setc 14 pd rec 15 12
pu fd 15 pd
repeat 2 [setc 13
          rec 15 12
          pu fd 15 pd]
pu bk 75 rt 90 fd 12 lt 90 pd
repeat 2 [setc 13
          rec 15 12
          pu fd 15 pd setc 14
          rec 15 12
          pu fd 15 pd]
setc 13 rec 15 12
pu bk 60
rt 90 pu fd 12 lt 90 pd
setc 14
rec 15 12
pu fd 15 pd

```

Draw three more rectangles,
changing
position after drawing each one.
Set the color to yellow.
Draw a rectangle
etc.,
etc.,
etc. . .

```

setc 13
repeat 3 [rec 15 12
  pu fd 15 pd]
setc 14 pd
rec 15 12
pu bk 60
rt 90 bk 48
lt 90 pd
end

```


```

to rec :a :b
repeat 2 [fd :a rt 90 Draw the rectangle.
  fd :b rt 90]
rt 45 pu fd :b      Move into position for fill.
pd fill pu          Fill.
bk :b lt 45 pd      Move back to the corner of the rectangle.
end

```

Stamping Shapes

The textile from Madagascar shown on page 5 was drawn by Antoine Draughon by stamping turtle shapes. Using the LogoWriter shape editor, he created four shapes that look like this:



The Logo weaving process involves positioning the turtle, setting its shape and color, then stamping the chosen shape in the chosen color at that position.

The **startup** procedure clears the screen, hides the turtle, and sets the background color to blue/green (**setbg 7**). The **slowturtle** command causes the pattern to emerge more slowly than normal so you can watch it more easily. It takes a while, but as Antoine said, "weaving takes time."

Finally, **startup** calls the procedure **antoine**, which does most of the work.

```

to startup
  rg ht ct
  setbg 7
  slowturtle
  antoine
end

```

In **antoine**, **antl** (for "antoine left") draws the left side of the pattern while **antr** draws the right side. The center portion of the design is drawn by the remainder of **antoine**. These last eight lines could have been made into another subprocedure, possibly called **antc for** "antoine center."

The process involves placing the turtle at a position near the bottom of the screen and setting a shape and color. Then a column of nine copies of the shape are stamped, each in a different color. Each of the 16 colors has a number. The instruction **setc color + 1** changes the color to the next one in the sequence. **Setc 2** makes the turtle green. Then **setc color + 1** sets it to color number 3, which is purple. The next **setc color + 1** instruction sets the turtle to color 4, which is red, and so on.

The command **setsh** sets the shape to one of the four patterns used to make the design.

The procedures **antl** is annotated below. The rest of the program is similar.

The procedure **details** displays information about the design.

```

to antoine
  antl
  antr
  pu setpos [30 -80]
  setsh 74
  setc 2
  repeat 9[setc color + 1 pd stamp pu fd 20]
  pu setpos [-30 -80]
  setsh 74
  setc 2
  repeat 9[setc color + 1 pd stamp pu fd 20]
  pu setpos [0 -80]
  setsh 73
  setc 2
  repeat 9[setc color + 1 pd stamp pu fd 20]
end

```

to antl

pu setpos [-150 -80]

setsh 73

setc 2

repeat 9[setc color + 1 pd
stamp pu fd 20]

setpos [-130 -80]

setsh 71

setc 2

repeat 9[setc color + 1 pd
stamp pu fd 20]

pu setpos [-110 -80]

setc 2

repeat 9[setc color + 1 pd
stamp pu fd 20]

pu setpos [-90 -80]

setsh 72

setc 2

repeat 9[setc color + 1 pd
stamp pu fd 20]

pu setpos [-70 -80]

setsh 72

setc 1

repeat 9[setc color + 1 pd
stamp pu fd 20]

pu setpos [-50 -80]

setsh 72

setc 2

repeat 9[setc color + 1 pd
stamp pu fd 20]

end

Put the turtle in the lower left corner of the screen.

Set a shape.

Set a color.

Set a new color, stamp the shape and move forward into the next position. Repeat this eight more times.

Put the turtle near the bottom of the screen, 20 steps to the right of where it was before.

Set a different shape.

Set a color.

Draw another column.

Get in position for the third column.

Set a color.

Draw another column.

Get in position.

Set a shape.

Set a color.

Draw another column.

Get in position.

Set a shape.*

Set a color.

Draw another column.

Get in position.

Set a shape.*

Set a color.

Draw another column.

* This instruction actually doesn't change the shape because it already was set to 72 by a previous instruction.

```

to antr
pu setpos [150 -80]
setsh 73
setc 2
repeat 9[setc color + 1 pd stamp pu fd 20]
pu setpos [130 -80]
setsh 71
setc 2
repeat 9[setc color + 1 pd stamp pu fd 20]
pu setpos [110 -80]
setc 2
repeat 9[setc color + 1 pd stamp pu fd 20]
pu setpos [90 -80]
setsh 72
setc 2
repeat 9[setc color + 1 pd stamp pu fd 20]
pu fd 20
pu setpos [70 -80]
setsh 72
setc 1
repeat 9[setc color + 1 pd stamp pu fd 20]
pu fd 20
pu setpos [50 -80]
setsh 72
setc 2
repeat 9[setc color + 1 pd stamp pu fd 20]
end

```

```

to details

```

```

settc 12
repeat 4[pr []]
tab

```

Pr[detail of a silk textile, Merina, Madagascar. The float weave patterns are distinctive of 19th century cloth reserved for aristocratic use. The blue/green background suggests that this example was woven for royalty, these being royal colors.]

```

pr []

```

```

tab

```

```

pr[by: Antoine Draughon, 7606]

```

```

pr[]

```

```

tab

```

```

pr[Spring 1992]

```

```

top

```


```

end

```


Shading


The large drawing by Coedell Page of the Njave cloth from Sierra Leone shown on page 9 illustrates one additional Logo graphics technique: shading. **Shade** is used the same way as **fill**. (Look at the explanation of **fill** in the discussion of Michael Hailstock's program above.) The turtle is moved to a position within an enclosed region. But instead of filling the region with a solid color, shade fills it with repeated copies of whatever shape the turtle is wearing at the time. For example, we could draw and shade a square like this:




```
repeat 4 [fd 80 rt 90]
```



```
pu rt 45 fd 40
```



```
setsh 29
```



```
pd shade ht
```

Here is Coedell Page's program in full. The design is much larger than the others. It is made up of three LogoWriter screens. The procedure **p** draws **des1** and prints it with **printscreen**. Then the screen is cleared and **des2** is drawn and printed. The third screen is the same as the first.

Most of the design is created by using the turtle to draw lines, with **Stamp** and **shade** being used in some places, as well.

Appendix - Different Versions of Logo

The African Textile project was carried out using GS LogoWriter {GS LogoWriter programs will run without modification in LogoEnsemble and in MSDOS LogoWriter with EGA or VGA graphics}. The procedures and shapes were adapted to Macintosh LogoWriter for inclusion in this publication. The work can be done in other versions of LogoWriter and in most versions of Logo, but some modifications may be necessary.

Screen Colors and Resolution

The students' work took advantage of the 16 colors available in GS LogoWriter. Your version of Logo may have more colors or fewer. The GS LogoWriter screen is 320 turtle steps wide and 190 steps high. The same is true for MSDOS LogoWriter and LogoEnsemble. Logos for the Macintosh and most MSDOS versions have a finer resolution. Versions of Logo for the Apple II have a coarser resolution.

Fill, Shade, and Stamp

These features are not present in all versions of Logo. Also, when they are present, they may be implemented in a different way.

Logo Vocabulary

Instead of **rg** for reset graphics, you may need to use **clearscreen**, **cs**, or **draw**. Instead of **setc** for set color, you may need to use **pencolor** or **pc**. Setting the position of the turtle might be done with **setxy** instead of **setpos**.